Il s'agit ici de tracer le spectre (en valeurs absolues) du développement en série de Fourier d'une fonction f.
On le fera pour :

- f paire de période T définie sur $\left[0,\frac{T}{2}\right]$ par y dépendant de t.
  On l'appelera `FourierP`, puis, dans un second temps `FourierP2`.
- f impaire de période T définie sur $\left]0,\frac{T}{2}\right[$ par y dépendant de t.
  On l'appelera `FourierI`, puis dans un second temps `FourierI2`.

1 Spectre jusqu'au rang n

1.1 Ensembles

Maple connait la notion d'ensemble.

- `{ }` représente l'ensemble vide,
- `A union B` représente la réunion d'ensembles,
- `plot({[x1,y1],[x2,y2],...,[xn,yn]},style = point,symbol = box)` trace les points indiqués sur un graphe, chaque point étant représenté par un petit carré.

1.2 Spectre

Ecrire 2 procédures `FourierP` et `FourierI` qui dessinent le spectre absolu de f définie précedemment respectivement entre 0 et n et entre 1 et n.
Elles auront donc 3 paramètres :

- y qui est une expression de t,
- T la période et,
- n le rang où on arrète.

On n'oubliera pas les `evalf`...
On essayera :

- `FourierP` avec
  - $y = t$
  - $T = 2\pi$
  - $n = 10$
- `FourierI` avec
  - $y = 1$
  - $T = 2\pi$
  - $n = 40$

1.3 Corrigé

```
FourierP := proc(y, T, n)
local ListP, i, ω;
    ListP := {[0, abs(2 * evalf(int(y, t = 0..1/2 * T))/T)]} ;
    ω := evalf(2 * π/T) ;
    for i to n do
        ListP := ListP union {[i, abs(4 * evalf(int(y * cos(i * ω * t), t = 0..1/2 * T)/T))]}
    end do;
    plot(ListP, style = point, symbol = box)
end proc
```

```
FourierI := proc(y, T, n)
local ListP, i, ω;
    ListP := {} ;
    ω := evalf(2 * π/T) ;
    for i to n do
        ListP := ListP union {[i, abs(4 * evalf(int(y * sin(i * ω * t), t = 0..1/2 * T)/T))]}
    end do;
    plot(ListP, style = point, symbol = box)
end proc
```

Explication des variables utilisées :

- $ListP$ est l'ensemble des points, chaque point est une liste de 2 éléments, l'abscisse et l'ordonnée,
- i est l'indice de l'harmonique sur laquelle on travaille,
- ω est la pulsation et,
- T est la période.

Explication de la démarche :
$ListP$ est initialisé au premier point (correspondant à la valeur moyenne) dans le cas pair et est initialisé au vide dans le cas impair.
Ensuite, on boucle de 1 à n en complétant à chaque fois $ListP$ par réunion avec un ensemble qui contient le nouveau point.
On termine en traçant le graphe.

2 Ratio d'information

2.1 Ratio d'information

On appelle Ratio d'information porté par la somme partielle d'une série de Fourier jusqu'au rang n le rapport :

$$\frac{a_0^2 + \sum_{i=1}^{n} \frac{(a_i^2 + b_i^2)}{2}}{\frac{\int_0^{\frac{T}{2}} y^2(t)\mathrm{d}t}{\frac{T}{2}}}$$

Ce ratio tend vers 1 quand $n \to +\infty$, c'est le théorème de Parseval.

2.2 Spectre

Ecrire 2 procédures `FourierP2` et `FourierI2` qui dessinent le spectre absolu de f définie précedemment jusqu'à ce que le ratio d'information soit plus grand qu'un paramètre e.
Elles auront donc 3 paramètres :

- y qui est une expression de t,
- T la période et,
- e le ratio minimum d'information à atteindre.

On n'oubliera pas les `evalf`...
On essayera :

- `FourierP2` avec
  - $y = t$
  - $T = 2\pi$
  - $e = 0.9999$

- `FourierI` avec
  - $y = 1$
  - $T = 2\pi$
  - $e = 0.99$

2.3 Corrigé

```
FourierP2 := proc(y, T, e)
local ListP, i, ω, n2, a, s;
  a := abs(2 * evalf(int(y, t = 0..1/2 * T))/T);
  n2 := evalf(2 * int(y^2, t = 0..1/2 * T)/T);
  ListP := {[0, a]};
  ω := evalf(2 * π/T);
  s := a^2;
  for i while evalf(s/n2) < e do
    a := abs(4 * evalf(int(y * cos(i * ω * t), t = 0..1/2 * T)/T));
    s := s + 1/2 * a^2;
    ListP := ListP union {[i, a]}
  end do;
  plot(ListP, style = point, symbol = box)
end proc
```

```
FourierI2 := proc(y, T, e)
local ListP, i, ω, n2, a, s;
  n2 := evalf(2 * int(y^2, t = 0..1/2 * T)/T);
  ListP := {};
  ω := evalf(2 * π/T);
  s := 0;
  for i while evalf(s/n2) < e do
    a := abs(4 * evalf(int(y * sin(i * ω * t), t = 0..1/2 * T)/T));
    s := s + 1/2 * a^2;
    ListP := ListP union {[i, a]}
  end do;
  plot(ListP, style = point, symbol = box)
end proc
```

Explication des variables :

- e est la valeur du ration d'information qu'il faut dépasser ($e > 1$),
- a est la valeur absolue du coefficient de Fourier,
- $n2$ est le carré de la norme de y,
- s est la somme partielle de la série de Parseval.

Explication de la démarche :

- On initialise d'abord les variables,
- on constitue ensuite point par point la liste $ListP$, tout en mettant à jour la somme partielle s, ceci tant que $\dfrac{s}{n2} < e$,
- on termine en traçant le graphe.